

PS/2 Auxiliary Device Data Format and Timing

PS/2 port Interface between Auxiliary device using GMS87C1408 and Host(PC)

DESCRIPTION

This document architects a class of auxiliary pointing devices which includes an extension of the PS/2™ Mouse as specified in the Mouse Technical Reference(IBM™ Pub. No. 68X2229). Typically, this includes such devices as trackballs, scratch pads and joystick. It also includes devices like TrackPoint II which allow the PS/2 Auxiliary Device Port to be used by two or more such devices, either effectively simultaneously or alternately under program control.

Here, the protocol between auxiliary device and host(PC) is only described, that is, data format and timing are shown in the text and figure.

And the PS/2 communication and operation of auxiliary device such as mouse, trackball, scratch pads and joystick is described in next application note.

PS/2 PORT PIN ASSIGNMENT

The auxiliary device is connected to the system by a 6 pin

DATA FORMAT

The data transmission between auxiliary device and host(PC) is composed of 11bits. this includes one start bit,

connector. The following shows pin numbers and signals.




Figure 1. PS/2 Connector

eight data bits, one parity bit and one stop bit. The data format is shown in below Figure 2.




Figure 2. Data Format

PS/2 Auxiliary Device Data Format and Timing

TIMING CHART AND EXPLANATION

HOST(PC) receiving Data

- PC samples data while CLK is low.

The following describes the typical sequence of events when the Host(PC) is receiving data from the Auxiliary Device

1. The Auxiliary Device checks the CLK line. If the line is inactive(low), output from the Auxiliary Device is not allowed.
2. The Auxiliary Device checks the DATA line. If the line is inactive(low), the Auxiliary Device controller receives data from the Host.
3. The Auxiliary Device checks the CLK line periodically during the transmission at intervals not exceeding 100

microseconds. If the Auxiliary Device finds that the Host is holding the CLK line inactive(low), the byte transmission is terminated. The Host can terminate transmission anytime during the first 10 clock cycles.

4. A final check for terminated transmission is performed at least 5 microseconds after clock ten.
5. The Host can hold the CLK line inactive(low) to inhibit the next transmission.
6. The Host can set the DATA line inactive(low) if it has a byte to transmit to the Auxiliary Device. The DATA line is set inactive(low) when the start bit(always 0) is placed on the DATA line.
7. The Host raises the CLK line to allow the next transmission.




Figure 3. Auxiliary Device Sending Data Timings

Timing	Description	Spec.(Min/Max)
T1	Time from DATA transition to falling edge of CLK1	5/25us
T1A	Time from DATA transition to falling edge of CLK 2-11	5/25us
T2	Time from rising edge of CLK to DATA transition	5/T4-5us
T3	Duration of CLK inactive(low)	30/50us
T4	Duration of CLK active(high)	30/50us
T5	Time to Auxiliary Device inhibit after clock 11 to ensure the Auxiliary Device does not start another transmission	0/50us
TCR	Line contention checking interval	0/100us

Table 1: Auxiliary Device Transmitting Data Timings

PS/2 Auxiliary Device Data Format and Timing

Host Sending Data

- *Auxiliary Device samples data while CLK is high.*

The following describes the typical sequence of events when the Host(PC) is sending data to the Auxiliary Device

1. The Host checks for an Auxiliary Device transmission in process. If a transmission is in process and beyond the 10th clock, the Host must receive the data.
2. The Auxiliary Device checks the CLK line. If the line is inactive(low), an I/O operation is not allowed.
3. The Auxiliary Device checks the DATA line. If the line is inactive(low), the Host has data to transmit. The DATA line is set inactive(low) when the start bit(always 0) is placed on the DATA line.
4. The Auxiliary Device sets the CLK line inactive(low). The Host then places the first bit on the DATA line. Each time the Auxiliary Device sets the CLK line inactive

ive(after falling edge), the Host places the next bit on the DATA line until all bits are transmitted.

5. The Auxiliary Device samples the DATA line for each bit while the CLK line is active(high). Data must be stable within 1microsecond after the rising edge of the CLK line.
6. The Auxiliary Device checks for a positive-level stop bit after the 10th clock. If the DATA line is inactive(low), the Auxiliary Device continues to clock until the DATA line becomes active(high). Then the Auxiliary Device clocks the line-control bit and, at the next opportunity, sends a Resend command to the Host.
7. The Auxiliary Device pulls the DATA line inactive(low), producing the line-control bit.
8. The Host can pull the CLK line inactive(low), inhibiting the Auxiliary Device.




Figure 4. Auxiliary Device Receiving Data Timings (I)

Timing	Description	Spec.(Min/Max)
T6	Auxiliary Device response to Host "request to send"	30us/10ms
T7	Duration of CLK inactive(low)	30us/50us
T8	Duration of CLK active(high)	30us/50us
T9	Time from inactive to active CLK transition, used to time when the Auxiliary Device samples DATA	5/25us
T10	Time from falling edge of line control bit to falling edge of clock 11 CLK	30/50us
T11	Time from rising edge of clock 11 to rising edge of line control bit	0/50us
TCX	Line contention checking interval	0/100us

Table 2: Auxiliary Device Receiving Data Timings (II)

Simple Hardware

A typical hardware connection is shown in Figure 5.




Figure 5. Simple Hardware and Connection

Author: *Sungjae Hwang*
 MCU application team

PS/2 Auxiliary Device Data Format and Timing

Appendix A:

GMS800 series MICOM ASSEMBLER Fri May 11 17:59:03 2001
(PAGE 1)

```
1          PAGE      1000
2          ;=====;
3          ;= Copy Right(c) Hynix Semiconductor 2001          =;
4          ;= All Right Reserved.                            =;
5          ;=====;
6          ;= Title: PS/2 Mouse Protocol                      =;
7          ;= Subject: 1, Transmit data(MSEADATA) to PC through PS/2          =;
8          ;=          2. Get the mouse command from PC throuth PS/2          =;
9          ;=          and Store it to Memory(MSECOMMAND)      =;
10         ;= Description:                                     =;
11         ;= Keyboard and mouse are connected to Personal Computer through PS/2 =;
12         ;= Port. The communication through PS/2 port is similar to that of I2C. =;
13         ;= That is, It is use two wires, one is clock and The other is data line.;
14         ;= This program is referred to IBM PS/2 Mouse specification(Ref.II,III) =;
15         ;=====;
16         ;= Device : GMS87C1408                              =;
17         ;= OSC      : 8MHz                                  =;
18         ;= Start Date: 2001, 5,                            =;
19         ;= End Date: 2001, 5,                              =;
20         ;= S/W      : SP/MCU Application Design Team        =;
21         ;= Developer: Sungjae Hwang                        =;
22         ;=====;
23         ;
24         ;Control Registers
25         RA      EQU      0C0H          ;[R/W] RA Port Data Reg.
26         RAIO    EQU      0C1H          ;[W] RA Port Direction Reg.
27         RAFUNC  EQU      0CAH          ;[W] RA Fuction Selection Reg.
28         IRQL    EQU      0E5H          ;[R/W] INT Request Reg. Low
29         WDTIF   EQU      6,0E5H
30         BITIF   EQU      5,0E5H
31         BITR    EQU      0ECH          ;[R] Basic Interval Timer Reg.
32         CKCTLR  EQU      0ECH          ;[W] Clock Control Register
33         WDTR    EQU      0EDH          ;[R],[W] Watchdog Timer Register
34
35
36         ;User Memory
37         MSEADATA DS      1             ;The data to transmit to PC
38         MSECOMMAND DS    1             ;The command to get from PC
39         COMMANDBUF DS    1             ;Store command from PC.
40         BITCNT   DS      1             ;To count the transmitted bit
41         PARITY   DS      1             ;To check parity
42         ;Constant
43         STACK    EQU      0BFH
44         OUTPUT_MODE EQU    1111_1111B ;MSESCL=RA.1,MSESDA=RA.0
45         INPUT_MODE EQU    1111_1100B ;MSESCL=RA.1,MSESDA=RA.0
46         SDA_INPUT_MODE EQU 1111_1110B ;MSESCL=RA.1,MSESDA=RA.0
47         ;Port Definition
48         MSESDA   EQU      0,RA         ;Mouse Data Definiton
49         MSESCL   EQU      1,RA         ;Mouse Clk Definition
50         ;Macro Definition
51         macro
52             bcc    \1                   ;A < #imm jump \1(label)
53             endm
54         ;=====;
55         ;= Vector Area                                     =;
56         ;=====;
57         ORG      0FFFEH
58 FFFE 00E0      DW      Reset          ; Reset
```

PS/2 Auxiliary Device Data Format and Timing

```

59      ;=====;
60      ;= Main Routine                                     =;
61      ;=====;
62      ORG      0E000H      ;GMS87C1408 Start Address
63      Reset:
64 E000 60      di
65 E001 40      clrgr
66 E002 1EBF    ldx      #STACK      ;Stack Point(00-BFh,OnlyExist 0-page,192byte)
67 E004 8E      txsps
68      ;
69 E005 1E00    ldx      #0
70      Clr0Page:
71 E007 C400    lda      #0
72 E009 FB      sta      {X}+
73 E00A 5EC0    cmpx     #STACK+1      ;0-Page RAM Clear(00-BFh)
74 E00C 70F9    bne      Clr0Page
75      ;
76 E00E E400C0  ldm      RA,#0000_0000B
77 E011 E4FCC1  ldm      RAO,#1111_1100B ;O RA7~2(Open,Output)
78      ;I RA1(PS2 Mouse CLK Input(Initial))
79      ;I RA0(PS2 Mouse Data Input(Initial))
80 E014 E400CA  ldm      RAFUNC,#0      ;Reset Value(RA Port I/O use)
81 E017 E41EEC  ldm      CKCTRL,#0001_1110B ;Initial(-001_0111B),BITCLK=1/(8Mhz/512)=64us
82 E01A E488ED  ldm      WDTR,#1000_1000B ;WDT(=64us*256*8=131.072ms)
83      ;No_useei
84      TrnsmitDataToPC:
85 E01D E4AA00  ldm      MSEDATA,#0AAh ;Mouse Diagnostics Success Tx Value
86 E020 3B43E0  call     !XmitMSE1ByteToPC
87 E023 3B16E1  call     !Delay500us
88 E026 E40000  ldm      MSEDATA,#00h ;Mouse ID
89 E029 3B43E0  call     !XmitMSE1ByteToPC
90 E02C 3B16E1  call     !delay500us
91      MSECCommandFromPC:
92 E02F 33C00A  bbc      MSESCL,MSECCommandEnd ;When MouseCLK(RA.1)=1 & MouseDATA(RA.0)=0,
93 E032 03C007  bbs      MSESDA,MSECCommandEnd ;The command from PC is exist.
94 E035 3BB6E0  call     !GetMSE1byteFromPC ;Stored MouseDataFromPC to MSECCommand
95 E038 C501    lda      MSECCommand
96 E03A E502    sta      COMMANDBUF ;Execute as Command
97      MSECCommandEnd:
98 E03C E488ED  ldm      WDTR,#1000_1000B ;WDT(=64us*256*8=131.072ms)
99 E03F B1E5    clr1     BITIF
100 E041 2FF9   bra      MSECCommandEnd
101      ;
102      ;=====;
103      ;= Sub Routine                                     =;
104      ;=====;
105      ;-----;
106      ;- Trnsmit Mouse_Data To Personal Computer      -;
107      ;-----;
108      XmitMSE1ByteToPC: ;Imachine(cpu)cycle:0.25us@8Mhz
109 E043 C500    lda      MSEDATA
110 E045 33C0FD  bbc      MSESCL,$ ;R0[R/W],R0DD[W,Write Only]
111 E048 13C0FD  bbc      MSESDA,$
112 E04B 21C0    set1     MSESCL
113 E04D 01C0    set1     MSESDA
114 E04F E4FFC1  ldm      RAO,#OUTPUT_MODE ;[w] Switch output mode
115      ; di ;MSESCL=RA.1,MSESDA=RA.0
116 E052 E40803  ldm      BITCNT,#8
117 E055 E40104  ldm      PARITY,#1 ;odd
118 E058 11C0    clr1     MSESDA ;I2C Start Bit("0")
119 E05A 21C0    set1     MSESCL
120      ;*** Time Fix ***

```

PS/2 Auxiliary Device Data Format and Timing

```

121 E05C 3B4AE1      call    !Delay19us    ;76cyc(19us,Total=80cycle(20us))
122 E05F 31C0        clr1   MSESCL        ;4cyc(1us)
123 E061 3B22E1      call    !Delay39us    ;156 cycle(39us,Total=160cycle(40us))
124 E064 21C0        set1   MSESCL        ;4cycle(1us,total=160cycle(40us))
125 E066 3B52E1      call    !Delay14us    ;56cycle(14us)
126
127 E069 68          XmitMSE8Bits:
128 E06A EBC010      stc    MSESDA        ;6cyc
129 E06D 5005        bcc    MSEParityAddZero ;2(No),4(Yes)
130 E06F 8904        inc    PARITY        ;4cyc
131 E071 FF          nop
132 E072 2F04        bra    MSEParityAddOne ;4cyc
133
134 E074 FF          MSEParityAddZero:
135 E075 FF          nop    ;2cyc
136 E076 FF          nop    ;2cyc
137 E077 FF          nop    ;2cyc
138
139 E078 21C0        MSEParityAddOne:
140 E07A 3B4AE1      set1   MSESCL        ;4cyc
141 E07D 31C0        call    !Delay19us    ;76cyc(19us,Total=80cycle(20us))
142 E07F 3B22E1      clr1   MSESCL        ;4cyc(1us)
143 E082 21C0        call    !Delay39us    ;156 cycle(39us,Total=160cycle(40us))
144 E084 3B50E1      set1   MSESCL        ;4cycle(1us,total=160cycle(40us))
145 E087 AC03DF      call    !Delay15us    ;61cyc(15.25us)
146
147 E08A 6904        dbne   bitcnt,XmitMSE8bits ;5(No),7(Yes)
148
149 E08C 6904        ;---
150 E08E EBC010      ror    PARITY        ;4cyc,to check odd or even(if data is odd then parity is 0)
151 E091 21C0        stc    MSESDA        ;6cyc;if C=1(data is odd) then set KBDDATALINE
152 E093 21C0        set1   MSESCL        ;4cyc,
153 E095 3B4AE1      call    !Delay19us    ;76cyc(19us,Total=80cycle(20us))
154 E097 31C0        clr1   MSESCL        ;4cyc(1us)
155 E099 3B22E1      call    !Delay39us    ;156 cycle(39us,Total=160cycle(40us))
156 E09B 21C0        set1   MSESCL        ;4cyc,
157 E09D 3B4CE1      call    !Delay18us    ;72cyc(18us,Total=80cycle(20us))
158
159 E09E 01C0        ;---
160 E0A0 21C0        set1   MSESCL        ;4cyc, I2C Stop Bit("1")
161 E0A2 3B4AE1      set1   MSESCL        ;4cyc,
162 E0A4 3B22E1      call    !Delay19us    ;76cyc(19us,Total=80cycle(20us))
163 E0A6 31C0        clr1   MSESCL        ;4cyc,
164 E0A8 3B22E1      call    !Delay39us    ;156 cycle(39us,Total=160cycle(40us))
165 E0AA 21C0        set1   MSESCL        ;4cyc
166
167 E0AB 21C0        ;*** Time Fix ***
168 E0AC E4FCC1      ; ei
169 E0AD 21C0        ldm    RAO,#INPUT_MODE ;[w] Switch input mode
170 E0AF 3B22E1      ;MSESCL=RA.1,MSESDA=RA.0
171 E0B1 3B22E1      call    !Delay39us    ;wait Ack(clk line)
172 E0B3 33C0FD      bbc    MSESCL,$      ;switched to High after 80us(Low)
173 E0B5 6F          ret
174
175 E0B7 6F          ;R0[R/W],RODD[W,Write Only]
176
177 E0B9 6F          ;-----;
178 E0BB 6F          ;- Get PC_Command_For_Mouse(Data) From PC and Store to RCVMSADATA-;
179 E0BD 6F          ;-----;
180
181 E0BF 6F          GetMSElbyteFromPC:
182 E0C1 33C0FD      bbc    MSESCL,$      ;MouseCLK(RA.1)=1 & MouseDATA(RA.0)=0;æ¹³
183 E0C3 03C0FD      bbs    MSESCL,$
184 E0C5 21C0        set1   MSESCL
185 E0C7 E4FEC1      ldm    RAO,#SDA_INPUT_MODE ;MouseClk=OutputMode
186 E0C9 E40803      ldm    BITCNT,#8
187 E0CB E40104      ldm    PARITY,#1
188 E0CD C400        lda    #0
189 E0CF E40001      ldm    MSECMMAND,#0
190
191 E0D1 6F          ;*** Time Fix
192 E0D3 31C0        clr1   MSESCL

```

PS/2 Auxiliary Device Data Format and Timing

```

183 E0CE 3B24E1      call    !Delay38us    ;152cyc
184 E0D1 FF          nop                                ;2cyc
185                  GetMSE8bits:
186 E0D2 FF          nop                                ;2cyc
187 E0D3 21C0        set1    MSESCL        ;4cyc
188 E0D5 CBC000      ldc     MSESDA        ;4cyc.(Bit0~Bit7)
189 E0D8 68          ror     A              ;2cyc.
190 E0D9 3B26E1      call    !Delay37us    ;148cyc
191 E0DC FF          nop                                ;2cyc.
192 E0DD 31C0        clr1    MSESCL        ;4cyc
193 E0DF 3B26E1      call    !Delay37us    ;147cyc.(148cyc)
194 E0E2 AC03ED      dbne   BITCNT,GetMSE8bits    ;5cyc(No)/7cyc(Yes)
195 E0E5 E501        sta     MSECOMMAND   ;4cyc.
196 E0E7 21C0        set1    MSESCL        ;4cyc.Parity Bit(10th Clk)
197 E0E9 3B22E1      call    !Delay39us    ;156cyc.9th_bit
198 E0EC 31C0        clr1    MSESCL        ;4cyc.
199 E0EE 3B22E1      call    !Delay39us    ;156cyc.
200 E0F1 21C0        set1    MSESCL        ;4cyc.
201 E0F3 3B4CE1      call    !Delay18us    ;72cyc.,10th_bit
202 E0F6 11C0        clr1    MSESDA        ;4cyc.Low Setting
203 E0F8 E4FCC1      ldm     RAI0,#INPUT_MODE    ;5cyc.Change Output(Clk,Data)
204 E0FB 11C0        clr1    MSESDA        ;4cyc.Low Setting
205 E0FD 3B4CE1      call    !Delay18us    ;71cyc.(72cyc)
206 E100 31C0        clr1    MSESCL        ;4cyc.
207 E102 3B22E1      call    !Delay39us    ;156cyc.
208 E105 21C0        set1    MSESCL        ;4cyc.
209 E107 3B4AE1      call    !Delay19us    ;76us
210 E10A 01C0        set1    MSESDA        ;4us
211                  ;*** Time Fix
212 E10C E4FCC1      ldm     RAI0,#INPUT_MODE
213 E10F 3B4AE1      call    !Delay19us
214 E112 33C0FD      bbc     MSESCL,$
215 E115 6F          ret
216                  ;-----;
217                  ;- Time Delay Routine                                -;
218                  ;-----;
219                  Delay500us:
220 E116 E41EEC      ldm     CKCTLR,#0001_1110B    ;Bit6(WAKEUP),Bit5(PCWDT),Bit4(WDTON)
221                  ;Bit3(BTCL),B210(110b=Fxin/512=64us)
222 E119 FF          nop                                ;64us*256=16.384mS at 8MHZ
223 E11A FF          nop
224                  Judge500us:
225 E11B C5EC        lda     BITR
226 E11D 4407        cmp     #07              ;64us*7=448us
227                  ble Judge500us
228 E11F 50FA        @    bcc     Judge500us
229 E121 6F          ret
230                  ;-----;
231                  Delay39us:
232 E122 FF          nop                                ;(39us)
233 E123 FF          nop
234                  Delay38us:
235 E124 FF          nop                                ;(38us)
236 E125 FF          nop
237                  Delay37us:
238 E126 FF          nop                                ;(37us)
239 E127 FF          nop
240 E128 FF          nop                                ;(36us)
241 E129 FF          nop
242 E12A FF          nop                                ;(35us)
243 E12B FF          nop
244 E12C FF          nop                                ;(34us)

```

PS/2 Auxiliary Device Data Format and Timing

245	E12D	FF	nop	
246	E12E	FF	nop	; (33us)
247	E12F	FF	nop	
248	E130	FF	nop	; (32us)
249	E131	FF	nop	
250	E132	FF	nop	; (31us)
251	E133	FF	nop	
252	E134	FF	nop	; (30us)
253	E135	FF	nop	
254	E136	FF	nop	; (29us)
255	E137	FF	nop	
256	E138	FF	nop	; (28us)
257	E139	FF	nop	
258	E13A	FF	nop	; (27us)
259	E13B	FF	nop	
260	E13C	FF	nop	; (26us)
261	E13D	FF	nop	
262	E13E	FF	nop	; (25us)
263	E13F	FF	nop	
264	E140	FF	nop	; (24us)
265	E141	FF	nop	
266	E142	FF	nop	; (23us)
267	E143	FF	nop	
268	E144	FF	nop	; (22us)
269	E145	FF	nop	
270	E146	FF	nop	; (21us)
271	E147	FF	nop	
272	E148	FF	nop	; (20us)
273	E149	FF	nop	
274			Delay19us:	
275	E14A	FF	nop	; (19us)
276	E14B	FF	nop	
277			Delay18us:	
278	E14C	FF	nop	; (18us)
279	E14D	FF	nop	
280	E14E	FF	nop	; (16us)
281	E14F	FF	nop	
282			Delay15us:	
283	E150	FF	nop	; (15us)
284	E151	FF	nop	
285			Delay14us:	
286	E152	FF	nop	; (14us)
287	E153	FF	nop	
288	E154	FF	nop	; (13us, call=8cycle)
289	E155	FF	nop	
290	E156	FF	nop	; (12us)
291	E157	FF	nop	
292	E158	FF	nop	; (11us)
293	E159	FF	nop	
294	E15A	FF	nop	; (10us)
295	E15B	FF	nop	
296	E15C	FF	nop	; (9us)
297	E15D	FF	nop	
298	E15E	FF	nop	; (8us)
299	E15F	FF	nop	
300	E160	FF	nop	; (7us)
301	E161	FF	nop	
302	E162	FF	nop	; (6us)
303	E163	FF	nop	
304	E164	FF	nop	; (5us)
305	E165	FF	nop	
306	E166	FF	nop	; 8cyc(4us, call=8cycle)(add+1)

PS/2 Auxiliary Device Data Format and Timing

```
307 E167 6F          ret          ; 5cyc
308          ;-----;
309          END
```

-- 0 Error(s) --

--- Total Machine Code : 362 Bytes ---

NOTE: